query: SELECT
    ?item ?articlename ?itemLabel ?itemDescription ?sl ?dob
    ?dod ?age ?dod_refs ?cod ?mod
  WITH
  {
    SELECT
      ?item ?dod_stm ?dod
      (GROUP_CONCAT(CONCAT(?dod_ref_propLabel, ": ", ?dod_ref_valueLabel); SEPARATOR=" ~ ") AS ?dod_refs)
    WHERE
    {
      VALUES ?dod {"+1953-03-05"^^xsd:dateTime}
      ?dod ^wdt:P570 ?item .
      ?item wdt:P31 wd:Q5 .
      OPTIONAL
      {
        ?item p:P570 ?dod_stm .
        ?dod_stm ps:P570 ?dod ;
                 psv:P570 [wikibase:timePrecision ?dod_precision].
        FILTER( ?dod_precision >= "11"^^xsd:integer )
        OPTIONAL
        {
          ?dod_stm prov:wasDerivedFrom ?dod_ref .
          ?dod_ref ?dod_ref_pr ?dod_ref_value .
          ?dod_ref_prop wikibase:reference ?dod_ref_pr .
        }
      }
      SERVICE wikibase:label
      {
        bd:serviceParam wikibase:language "en" .
        ?dod_ref_prop rdfs:label ?dod_ref_propLabel .
        ?dod_ref_value rdfs:label ?dod_ref_valueLabel .
      }
    }
    GROUP BY ?item ?dod_stm ?dod
  } AS %get_dod_refs
  WITH
  {
    SELECT
      ?item ?dod_stm ?dod ?dod_refs (GROUP_CONCAT(?codLabel; SEPARATOR=", ") AS ?cod)
    WHERE
    {
      INCLUDE %get_dod_refs
      OPTIONAL
      {
        ?item p:P509 ?cod_stm .
        ?cod_stm ps:P509 ?cod .
      }
      SERVICE wikibase:label
      {
        bd:serviceParam wikibase:language "en" .
        ?cod rdfs:label ?codLabel .
      }
    }
    GROUP BY ?item ?dod_stm ?dod ?dod_refs
  } AS %get_cod
  WITH
  {
    SELECT
      ?item ?dod_stm ?dod ?dod_refs ?cod (SAMPLE(?modLabel) AS ?mod)
    WHERE
    {
      INCLUDE %get_cod
      OPTIONAL
      {
        ?item p:P1196 ?mod_stm .
        ?mod_stm ps:P1196 ?mod .
      }
      SERVICE wikibase:label
      {
        bd:serviceParam wikibase:language "en" .
        ?mod rdfs:label ?modLabel .
      }
    }
    GROUP BY ?item ?dod_stm ?dod ?dod_refs ?cod
  } AS %get_mod
  WHERE
  {
    INCLUDE %get_mod
    ?item wikibase:sitelinks ?sl .
    ?item ^schema:about ?article .
    ?article schema:isPartOf <https://en.wikipedia.org/> ;
             schema:name ?articlename .
    OPTIONAL { ?item wdt:P569 ?dob ;
                     p:P569 ?dob_stm .
               ?dob_stm ps:P569 ?dob ;
                        psv:P569 [wikibase:timePrecision ?dob_precision].}
    SERVICE wikibase:label
      {
        bd:serviceParam wikibase:language "en" .
        ?item rdfs:label ?itemLabel .
        ?item schema:description ?itemDescription .
      }
    BIND(REPLACE(?itemLabel, "^.*(?<! [Vv][ao]n| [Dd][aeiu]| [Dd][e][lns]| [Ll][ae]) (?!([SJ]r\\.?|[XVI]+)$)", "") AS ?sortname)
    BIND(IF(BOUND(?dob),IF(?dob_precision>=11,str(if(?dod< strdt(concat(str(year(?dod)),"-",if(strlen(str(month(?dob)))=1,"0",""),str(month(?dob)),"-",if(strlen(str(day(?dob)))=1,"0",""),str(DAY(?dob))),xsd:dateTime),year(?dod)-year(?dob)-1,year(?dod)-year(?dob))),"exact DOB unknown"),"DOB missing") AS ?age)
  } ORDER BY ASC(UCASE(?sortname)) ASC(UCASE(?itemLabel))